

9/10.3

Carnet de rendez-vous

Ce programme écrit en Basic est un outil qui peut vous rendre de grands services. Il s'agit d'un carnet de rendez-vous électronique. Ses fonctions sont les suivantes :

- calendrier mensuel automatique, pour une année comprise entre 1900 et 1999 ;
- emploi du temps : possibilité de noter sous forme texte vos rendez-vous de la journée, de 8 h 00 à 20 h 00 ;
- recherche d'un rendez-vous par mot clé. Le ou les jours qui comportent le mot clé s'affichent sur l'écran.

Ce programme utilise le fichier HCBIN.BIN donné avec le programme de hard-copy graphique dans un complément précédent (Partie 9, Chapitre 8.1). Veillez à placer ce fichier sur la même disquette que le programme Basic « Carnet de rendez-vous » ou à insérer la disquette qui le contient avant de lancer l'exécution. Nous rappelons que ce fichier peut être copié en utilisant l'utilitaire CP/M FILECOPYY.

COMMENT EXÉCUTER LE PROGRAMME

Lancez le programme. Après deux accès disques (chargement du fichier HCBIN.BIN et de l'éventuel fichier RDV qui contient les rendez-vous), le programme demande d'entrer la date d'aujourd'hui sous la forme jj/mm/aa. Entrez par exemple 10/09/89 pour le 10 septembre 1989. Après un bref calcul, le carnet de rendez-vous s'affiche sur l'écran, sous forme graphique, comme le montre la copie d'écran ci-dessous :

Calendrier Dimanche 10 Septembre 1989						
Dim	Lun	Mar	Mer	Jeu	Ven	Sam
					1	2
3	4	5	6	7	8	9
10	11	12	13	14	15	16
17	18	19	20	21	22	23
24	25	26	27	28	29	30

En dessous du calendrier, un menu à sept options apparaît. Elles doivent être utilisées comme suit :

— Jour suivant : cette option augmente d'un le jour courant. Si le jour courant est le dimanche 10 septembre 1989, l'appui sur la touche 1 fait afficher lundi 11 septembre 1989. Si nécessaire, le calendrier du mois suivant est affiché.

— Jour précédent : cette option diminue d'un le jour courant. Si le jour courant est le dimanche 10 septembre 1989, l'appui sur la touche 2 fait afficher samedi 9 septembre 1989. Si nécessaire, le calendrier du mois précédent est affiché.

— Editer date : lorsque la date que l'on désire afficher est trop éloignée de la date courante, cette option doit être utilisée. Le format de la date entrée est le même que celui demandé en début de programme : jj/mm/aa.

— Impression : cette option permet d'imprimer sous forme graphique le calendrier. L'image obtenue est similaire à celle donnée en exemple ci-dessus.

— Emploi du temps : cette option donne accès à l'emploi du temps. Lorsque vous la sélectionnez, le calendrier s'efface pour laisser la place à un écran sur lequel figurent des indications d'heures 8 h, 9 h, 10 h, ..., 20 h. Un point d'interrogation en face de la ligne marquée 8 h signale que l'ordinateur attend du texte à cette position. Appuyez sur la touche **Enter** pour passer à la ligne suivante. Lorsque vous êtes arrivé en face de l'heure qui vous intéresse, notez une information. Par exemple, **Repas avec M. Vidal**, ou encore **Réunion d'avancement**.

Lorsque vous resélectionnez l'emploi du temps de cette même journée, les informations que vous avez notées seront fidèlement restituées. Vous pourrez ainsi les voir, ou les modifier...

— Recherche : cette option vous permet de rechercher une chaîne texte dans l'emploi du temps. Tous les champs de l'emploi du temps sont parcourus, et la ou les éventuelles occurrences sont signalées sous la forme "**Chaîne trouvée le jj/mm/aa**". Dans le cas où la chaîne spécifiée n'existe pas dans l'emploi du temps, le message "**Chaîne inexistante dans l'emploi du temps**" est affiché.

— Fin : cette option met fin au programme et sauvegarde les données de l'emploi du temps dans le fichier EDT. La capacité maximale de ce fichier est de 50 enregistrements. Nous verrons dans un prochain complément comment augmenter cette capacité en gérant les fiches obsolètes.

LE PROGRAMME EN DÉTAIL

Au début du programme, les tableaux de travail sont dimensionnés :

```
DIM m$(12),nj(12),t(42),rdv$(50,14)
```

Le programme binaire de hard-copy graphique est ensuite chargé :

```
MEMORY &9000 : LOAD "hcbn"
```

Lorsque le fichier RDV existe, les données qu'il contient sont chargées en mémoire dans une boucle FOR NEXT :

```
FOR j = 1 to 14
  INPUT #9,rdv$(ne,j)
NEXT j
```

Les données du programme sont ensuite mémorisées dans leurs tableaux respectifs à l'aide d'une série de boucles FOR NEXT (jours de la semaine, mois et nombre de jours de chaque mois).

Après la saisie de la date courante, la procédure située ligne 3440 est appelée. Cette procédure calcule le jour correspondant au premier du mois et stocke le résultat dans la variable z.

Le calendrier est affiché sous forme semi-graphique, à l'aide de plusieurs codes ASCII compris entre 147 et 159. Ces codes sont stockés dans des variables pour être plus facilement manipulés :

```
c1$ = CHR$(150)
```

```
...
```

```
C0$ = CHR$(155)
```


Après divers calculs graphiques, le calendrier est affiché à l'aide d'une série de PRINT, et le menu général est affiché :

```
2220 PRINT "    1 = Jour suivant, 2 = Jour précédent, 3-Editer
date, 4 = Impression"
2230 PRINT "    5 = Emploi du temps, 6 = Recherche, 7-Fin"
```


Le choix de l'utilisateur est acquis dans la variable a\$ à l'aide d'une boucle sur l'instruction INKEY\$:

```
2240 a$ = INKEY$;IF a$ = "" THEN 2240
2250 ON ASC(a$) - 48 GOTO 2300,2500,1470,2710,2860,
3110,3290
```

L'appui sur la **touche 1** donne accès au traitement qui incrémente le jour courant. La logique de ce traitement apparaît dans l'ordinogramme suivant :



L'appui sur la **touche 2** donne accès au traitement qui décrémente le jour courant. La logique de ce traitement est relativement similaire à celle du précédent. En voici le détail :



L'appui sur la **touche 3** réexécute le programme à partir de la ligne 1470, ce qui a pour effet de redemander à l'utilisateur la date courante.

L'appui sur la **touche 4** lance l'impression graphique. Reportez-vous au programme de hard-copy graphique pour avoir tous les détails sur son fonctionnement (voir Partie 9, chapitre 8.1).

L'appui sur la **touche 5** active le traitement Emploi du temps. Ce traitement découpe la journée courante en 12 tranches horaires de une heure, entre 8 heures et 20 heures. Lorsque le tableau enr\$ contient des informations sur la journée courante, elles sont affichées sur les lignes correspondantes. Dans tous les cas, l'utilisateur peut rajouter ou modifier les données affichées sur l'écran. L'appui sur la touche **Enter** fait passer au champ suivant sans effacer le calcul du champ courant.

L'appui sur la **touche 6** donne accès au programme de recherche. Ce programme recherche une chaîne dans tous les champs du tableau rdv\$, c'est-à-dire dans toutes les fiches de l'emploi du temps. Cette recherche est de type EXACT, c'est-à-dire que la chaîne tapée par l'utilisateur est exactement recherchée dans chaque champ. Une forme similaire n'est jamais reconnue. Nous verrons comment améliorer ce type de recherche dans un prochain complément.

Enfin, l'appui sur la **touche 7** sauvegarde l'emploi du temps dans le fichier RDV :

```
FOR i=1 to ne
  FOR j=1 to 14
    PRINT #9,rdv$(i,j)
  NEXT j
NEXT i
```

Le listing du programme :

```
1000 REM =====
1010 REM Carnet de rendez-vous
1020 REM =====
1030 '
1040 DIM m$(12),nj(12),t(42),rdv$(50,14)
1050 WIDTH 255 'Pas de retour charriot automatique
1060 MEMORY &9000 : LOAD "hcbn" 'Chargement du S/P Assemble
ur
1070 '
1080 FOR i=1 TO 50
1090   FOR j=1 TO 14
1100     rdv$(i,j)=" "
1110   NEXT j
1120 NEXT i
1130 '
1140 ne=0 'Nombre d'enregistrements
1150 ON ERROR GOTO 3920
1160 OPENIN "RDV"
```

```

1170 '
1180 ne=1
1190 FOR j=1 TO 14
1200   INPUT #9,rdv$(ne,j)
1210 NEXT j
1220 IF (EOF=0) AND (ne<>50) THEN ne=ne+1:GOTO 1190
1230 CLOSEIN
1240 '
1250 '-----
1260 ' Lecture des donnees
1270 '-----
1280 '
1290 FOR i=1 TO 7
1300   READ j$(i)
1310 NEXT i
1320 FOR i=1 TO 12
1330   READ m$(i)
1340 NEXT i
1350 FOR i=1 TO 12
1360   READ nj(i)
1370 NEXT i
1380 '-----
-----
1390 DATA Dimanche,Lundi,Mardi,Mercredi,Jeudi,Vendredi,Samed
i
1400 DATA Janvier,Fevrier,Mars,Avril,Mai,Juin
1410 DATA Juillet,Aout,Septembre,Octobre,Novembre,Decembre
1420 DATA 31,28,31,30,31,30,31,31,30,31,30,31
1430 '-----
-----
1440 CLS
1450 PRINT SPACE$(27);"CARNET DE RENDEZ-VOUS"
1460 PRINT SPACE$(27);"=====
1470 LOCATE 1,19
1480 enr=0
1490 INPUT "Entrez la date d'aujourd'hui sous la forme jj/mm
/aa : ";d$
1500 CLS
1510 jc=VAL(LEFT$(d$,2))
1520 jo=1
1530 mo=VAL(MID$(d$,4,2))
1540 an$="19"+RIGHT$(d$,2)
1550 GOSUB 3440 'Calcul du jour correspondant au premier du
mois
1560 '-----
-----
1570 '
1580 '-----
1590 ' Caracteres graphiques
1600 '-----
1610 '

```

```

1620 c1#=CHR$(150)
1630 c2#=CHR$(156)
1640 c3#=CHR$(147)
1650 c4#=CHR$(153)
1660 c5#=CHR$(149)
1670 c6#=CHR$(151)
1680 c7#=CHR$(157)
1690 c8#=CHR$(158)
1700 c9#=CHR$(159)
1710 c0#=CHR$(155)
1720 '
1730 '-----
1740 ' Calcul des informations a afficher
1750 '-----
1760 '
1770 a$(1)=c1#+STRING$(15,154)+"Calendrier"+STRING$(16,154)+
c2#
1780 ax#=STRING$(5,154)+c8#
1790 a$(3)=c6#+ax#+ax#+ax#+ax#+ax#+ax#+STRING$(5,154)+c7#
1800 a$(4)=c5#+ " Dim "+c5#+ " Lun "+c5#+ " Mar "+c5#+ " Mer "
1810 a$(4)=a$(4)+c5#+ " Jeu "+c5#+ " Ven "+c5#+ " Sam "+c5#
1820 ax#=STRING$(5,154)+c9#
1830 a$(5)=c6#+ax#+ax#+ax#+ax#+ax#+ax#+STRING$(5,154)+c7#

1840 FOR i=1 TO 42
1850   t(i)=0
1860 NEXT i
1870 FOR i=1 TO z
1880   t(i)=0
1890 NEXT i
1900 FOR i=z+1 TO z+nj(mo)
1910   t(i)=i-z
1920 NEXT i
1930 ax#=STRING$(5,154)+c0#
1940 a$(6)=c3#+ax#+ax#+ax#+ax#+ax#+ax#+STRING$(5,154)+c4#
1950 jo=jc
1960 GOSUB 3440 'Calcul du jour correspondant a la date cour
ante
1970 a$(2)=c5#+SPACE$(15-LEN(j$(z+1)))+j$(z+1)+" "
1980 a$(2)=a$(2)+SPACE$(3-LEN(STR$(jo)))+STR$(jo)
1990 a$(2)=a$(2)+SPACE$(10-LEN(m$(mo)))+m$(mo)+" "+an#
2000 a$(2)=a$(2)+"          "+c5#
2010 '
2020 '-----
2030 ' Affichage du calendrier
2040 '-----
2050 '
2060 CLS
2070 FOR i=1 TO 5
2080   PRINT SPACE$(15)+a$(i)
2090 NEXT i
2100 FOR i=1 TO 6

```

```

2110 PRINT SPACE$(15);
2120 FOR j=1 TO 7
2130   a$=STR$(t((i-1)*7+j))
2140   IF t((i-1)*7+j)=0 THEN PRINT c5$+"      ";;GOTO 2160
2150   PRINT c5$+SPACE$(3-LEN(a$))+a$+" ";
2160 NEXT j
2170 PRINT c5$;
2180 PRINT
2190 NEXT i
2200 PRINT SPACE$(15)+a$(6)
2210 LOCATE 1,16
2220 PRINT"          1=Jour suivant, 2=Jour precedent, 3=Editer
data, 4=Impression"
2230 PRINT"                    5=Emploi du temps, 6=Recherche
, 7=Fin"
2240 a$=INKEY$; IF a$="" THEN 2240
2250 ON ASC(a$)-48 GOTO 2300,2500,1470,2710,2860,3110,3290
2260 GOTO 2240
2270 END
2280 '
2290 REM -----
2300 REM Jour suivant
2310 REM -----
2320 '
2330 jo=jo+1;enr=0
2340 GOSUB 3440 'Calcul du jour
2350 a$(2)=c5$+SPACE$(15-LEN(j$(z+1)))+j$(z+1)+" "
2360 a$(2)=a$(2)+SPACE$(3-LEN(STR$(jo)))+STR$(jo)
2370 a$(2)=a$(2)+SPACE$(10-LEN(m$(mo)))+m$(mo)+" "+an$
2380 a$(2)=a$(2)+"          "+c5$
2390 IF jo>nj(mo) THEN GOTO 2440
2400 LOCATE 16,2
2410 PRINT a$(2)
2420 GOSUB 3950 'Calcul de d$
2430 GOTO 2240
2440 mo=mo+1;jo=1;jc=1
2450 IF mo=13 THEN mo=1;an$="19"+RIGHT$(STR$(VAL(an$)+1),2)
2460 GOSUB 3950 'Calcul de d$
2470 GOTO 1550
2480 '
2490 REM -----
2500 REM Jour precedent
2510 REM -----
2520 '
2530 jo=jo-1;enr=0
2540 GOSUB 3440 'Calcul du jour
2550 a$(2)=c5$+SPACE$(15-LEN(j$(z+1)))+j$(z+1)+" "
2560 a$(2)=a$(2)+SPACE$(3-LEN(STR$(jo)))+STR$(jo)
2570 a$(2)=a$(2)+SPACE$(10-LEN(m$(mo)))+m$(mo)+" "+an$
2580 a$(2)=a$(2)+"          "+c5$
2590 IF jo=0 THEN GOTO 2640

```

```
2600 LOCATE 16,2
2610 PRINT a$(2)
2620 GOSUB 3950 'Calcul de d$
2630 GOTO 2240
2640 mo=mo-1
2650 IF mo=0 THEN mo=12:an$="19"+RIGHT$(STR$(VAL(an$)-1),2)
2660 jc=nj(mo):jo=1
2670 GOSUB 3950 'Calcul de d$
2680 GOTO 1550
2690 '
2700 REM -----
2710 REM Impression
2720 REM -----
2730 '
2740 LOCATE 1,16
2750 PRINT SPACE$(78)
2760 LOCATE 1,17
2770 PRINT SPACE$(78)
2780 GOTO 3560
2790 LOCATE 1,16
2800 PRINT"          1=Jour suivant, 2=Jour precedent, 3=Editer
date, 4=Impression"
2820 PRINT"                      5=Emploi du temps, 6=Recherche
, 7=Fin"
2830 GOTO 2240
2840 '
2850 REM -----
2860 REM Emploi du temps
2870 REM -----
2880 '
2890 CLS
2900 PRINT SPACE$(30)+"Emploi du temps"
2910 PRINT SPACE$(30)+"======"
2920 PRINT:PRINT
2930 FOR i=1 TO ne
2940   IF rdv$(i,1)=d$ THEN enr=i 'No enregistrement
2950 NEXT i
2960 '
2970 FOR i=1 TO 13
2980   IF (i+7)<10 THEN PRINT" ";
2990   PRINT i+7;"h : ";
3000   IF enr<>0 THEN PRINT SPACE$(3);rdv$(enr,i+1) ELSE PRI
NT
3010 NEXT i
3020 IF enr=0 THEN enr=ne+1:rdv$(enr,1)=d$:ne=ne+1
3030 FOR i=1 TO 13
3040   LOCATE 10,i+4
3050   INPUT a$
3060   IF a$<>"" THEN rdv$(enr,i+1)=a$
3070 NEXT i
3080 GOTO 1550
```

```

3090 '
3100 REM -----
3110 REM Recherche
3120 REM -----
3130 '
3140 CLS
3150 PRINT SPACE$(25);"Recherche d'une chaine dans l'emploi
du temps"
3160 PRINT SPACE$(25);"=====
=====."
3170 PRINT:PRINT
3180 INPUT "Chaine recherchee : ";cr$
3190 tr=0
3200 FOR i=1 TO ne
3210   FOR j=1 TO 14
3220     IF rdv$(i,j)=cr$ THEN tr=1:PRINT "Chaine trouvee le
";rdv$(i,1)
3230   NEXT j
3240 NEXT i
3250 IF tr=0 THEN PRINT"Chaine inexistante dans l'emploi du
temps."
3260 PRINT:PRINT "Appuyez sur une touche..."
3270 a$=INKEY$:IF a$="" THEN 3270
3271 GOTO 1550
3280 '
3290 REM -----
3300 REM Sauvegarde de l'emploi du temps
3310 REM -----
3320 '
3330 OPENOUT "rdv"
3340 FOR i=1 TO ne
3350   FOR j=1 TO 14
3360     PRINT #9,rdv$(i,j)
3370   NEXT j
3380 NEXT i
3390 CLS
3400 END
3410 GOTO 1550
3420 END
3430 '
3440 REM -----
3450 REM Calcul d'un jour de l'annee
3460 REM -----
3470 '
3480 IF INT(VAL(an$)/4)*4 = VAL(an$) THEN nj(2)=29 ELSE nj(2)
)=28 'Bisextile
3490 m=mo-2:IF mo<=2 THEN m=mo+10
3500 c=VAL(LEFT$(an$,2)):a=VAL(RIGHT$(an$,2))
3510 IF mo<=2 THEN a=a-1
3520 x=INT(2.6*m-0.199)+jo+a+INT(a/4)+INT(c/4)-2*c
3530 z=INT(x)-7*INT(x/7)

```

```

3540 RETURN
3550 '
3560 '-----
3570 ' Hard-Copy graphique rapide DMP 2000
3580 '-----
3590 '-----
3600 ' Initialisation
3610 '-----
3620 PRINT#8,CHR$(27);"@";PRINT#8,CHR$(27);"1"
3630 ORIGIN 0,0 : ve%=401
3640 '-----
3650 ' Programme principal
3660 '-----
3670 a=@ve% : amsb=PEEK(a+1) : alsb=PEEK(a) 'Interface avec
le S/P Assembleur
3680 POKE &905C,alsb : POKE &905D,amsb
3690 CALL &9000 'Calcul d'une ligne graphique
3700 GOSUB 3780 'Trace
3710 ve%=ve%-14
3720 IF ve%>0 THEN 3670 'Passage a la prochaine colonne
3730 GOTO 2800
3740 '-----
3750 ' Sous-programme d'impression
3760 '-----
3770 '
3780 a=0:af=0:a(0)=1:a(1)=100:a(2)=100:a(3)=120
3790 FOR j=1 TO 3
3800   a=a+a(j-1):af=af+a(j)
3810   PRINT#8,CHR$(27);"K";CHR$(a(j));CHR$(0);
3820   FOR k=a TO af
3830     PRINT#8,CHR$(PEEK(&9069+k-1));
3840   NEXT k
3850 NEXT j
3860 PRINT#8
3870 RETURN
3880 REM -----
3890 REM Gestion des erreurs
3900 REM -----
3910 '
3920 RESUME 1250
3930 '
3940 REM -----
3950 REM Calcul de d$
3960 REM -----
3970 '
3980 IF jo<10 THEN d$="0"+CHR$(48+jo)+"/" ELSE d$=RIGHT$(STR
$(jo),2)+"/"
3990 IF mo<10 THEN d$=d$+"0"+CHR$(48+mo)+"/" ELSE d$=d$+RIGH
T$(STR$(mo),2)+"/"
4000 d$=d$+RIGHT$(an$,2)
4010 RETURN

```